

## Using SAS® Macro to Include Statistics Output in Clinical Trial Summary Table

Amy C. Young, Ischemia Research and Education Foundation, San Francisco, CA  
Sharon X. Zhou, Ischemia Research and Education Foundation, San Francisco, CA

### ABSTRACT

The summary table is one of the common components of the clinical study reports. The table usually includes the statistics such as, counts, means, standard deviations, medians, ranges for continuous variables; number and percentage of patients falling within a particular category for categorical variables. The tasks for reporting these information are similar in the study; and therefore can be standardized using SAS Macro. This paper will show the users the development of SAS Macro that generates and reports the summary statistics of the treatment groups for either continuous or categorical efficacy variables, or both. The macro can be easily modified to include other summary statistics and for different numbers of treatment groups.

### INTRODUCTION

In clinical trials, data are presented in summary tables in statistical and clinical reports in order to facilitate the interpretation of the results, and to understand the data better before performing any treatment contrasts. Most summary reports display these features:

- Number and percentage of patients in each treatment group in whom one particular event occurred. They are reported most frequently for categorical variables.
- Computational data such as, mean, standard deviation, median, minimum, and maximum of each treatment group are often summarized for continuous variables.

This paper exhibits a SAS program that uses PROC FREQ, PROC UNIVARIATE, and ARRAY in a SAS Macro to generate the clinical summary report which includes both continuous and categorical variables. We attach the SAS programming codes used to produce the Cardioplegia measurements summary report in the Cardiopulmonary Bypass as an example to demonstrate the described methodology. The measurements contain both continuous (i.e. core temperature, amount of Cardioplegia) and categorical (i.e. Antegrade, Retrograde, Crystalloid) variables.

### SAS PROGRAM DEVELOPMENT

#### Preparation of Analysis Data Set

The first step before developing the macro to calculate the summary statistics is to prepare the analysis data set. This includes subsetting, sorting, merge, or any other necessary clean-up. If you intend to include the overall statistics for all treatments in the report, you would need to assign a dummy group code for the "Total" column. For an example:

```
DATA TEMP2;
SET TEMP1;
  BY ID;
  OUTPUT;
  GRPCODE =5; *-- Assign group code for overall statistics.
              *--Grpcode for total = total number of
              treatments + 1. Here, grpcode=5 since we
              have 4 treatments.;
  OUTPUT;
RUN;
```

In this study, we use numeric group code to identify different treatments. By doing so, we can then display the treatment group statistics in the desired order.

If you prefer to list the reporting variables statistics in a particular order, the other important task in the analysis data preparation is to make sure that the reporting variables are named in alphabetical order. This alphabetical order should correspond to the desired row order in the summary table. For instances, in our summary report for Cardiopulmonary Bypass, we would like to list the statistics in the row order of core temperature, amount of cardioplegia, antegrade, retrograde, and crystalloid; thus, we renamed these variables in alphabetical order as: CORETEMP, CPLEGIA, XAGRADE, XRGRADE, XTALLOI. This is because later in the report generation, the analysis data set will be sorted by the reporting variable names. If the variables are not named alphabetically to follow your chosen sequence, they will not be listed accordingly in the table. Therefore, rename the variables if necessary.

### Generating Summary Statistics

#### Step 1. Assign macro variables:

For each listed category in the summary report, we assign the following macro variables:

- variable names: this is the variable you wish to report in the summary table. In our example, these variables are: core temperature, amount of Cardioplegia, etc..
- variable type: identify whether the variable is continuous or categorical.
- variable format: if any format is used.
- variable label: the name of the reporting variable that will be displayed in the summary table.

```
%LET VAR1 = CORETEMP; *--reporting variable name;
%LET VARTYP1 = CON;   * type (CON = continuous,
                     CAT = categorical);
%LET VARFMT1 = ;     * format name – do not include "."
                     from the format.
%LET VARLAB1 = %STR(LOWEST CORE TEMPERATURE ON
                    CPB); * Variable Label;
```

VAR1 will be the first category displayed in the report, VAR2 the second, and so on. Thus, the macro variables should be allocated according to the desired display order

in the summary table as described in the analysis data set preparation.

## Step 2. Macro to generate summary statistics:

The macro developed here will report two major groups of statistics.

I. The total number of patients in each treatment groups: We use PROC FREQ procedure to obtain the total number of patients in each treatment group, and output the result to a data set. Then we apply PROC TRANSPOSE to arrange the output data (from PROC FREQ) for display.

II. The summary statistics: we apply different procedures to generate the summary statistics of each treatment group according to the type of the variables.

**Categorical** – We only report the number and the percentage of the patients for categorical variable. Thus, the method is similar as described at (I). We run PROC FREQ, and PROC TRANSPOSE procedures to acquire the number and percentage of patients in each treatment group, and to organize the results for the report.

**Continuous** – For continuous variables, the treatment group means, standard deviations, medians, the minimum, and the maximum are often reported instead of the number/percentage of patients in the summary table. We use PROC UNIVARIATE procedure to obtain these statistics. To arrange these statistical results by treatment groups, we run the ARRAY function to assign new variable names for the statistics according to its associated treatment group; i.e.:

```

ARRAY _MEAN{&TRTNUM} MEAN1-MEAN&TRTNUM;
ARRAY _MEDIAN{&TRTNUM} MEDIAN1-MEDIAN&TRTNUM;
ARRAY _MIN{&TRTNUM} MIN1-MIN&TRTNUM;
ARRAY _MAX{&TRTNUM} MAX1-MAX&TRTNUM;
ARRAY _STD{&TRTNUM} STD1-STD&TRTNUM;
ARRAY _NUM{&TRTNUM} CNT1-CNT&TRTNUM;
ARRAY _MISS{&TRTNUM} MISCNT1-MISCNT&TRTNUM;
.....
DO I=1 TO &TRTNUM;
  SET STATS END=LAST;
  _MEAN(I) = MEAN;
  _MEDIAN(I) = MEDIAN;
  _MIN(I) = MIN;
  _MAX(I) = MAX;
  _STD(I) = STD;
  _NUM(I) = COUNT;
  _MISS(I) = MISS;

  IF LAST THEN RETURN;
END;
```

Step 2 needs to be run for each macro variables you defined in step 1.

## Producing the summary table

To display the summary statistics in a report form, we build macros to assign the column width for each treatment group, and the column format (as shown in the attached SAS codes). Finally, a macro includes DATA \_NULL\_ and PUT function is constructed to display the statistics in the summary table.

You may also consider using PROC REPORT procedure as an alternative way to produce the summary table.

## SUMMARY

The demonstrated SAS Macro presents the ability to report both the categorical and continuous variables summary statistics on the same table. In addition to each individual treatment group statistics, it can also include the overall summary statistics in the "Total" column. This technique can be applied to demographic, medical history, and many other efficacy reports. The formatted output also increases the ability for statisticians or other reviewers to quickly understand the presented information.

## REFERENCE

SAS Institute Inc., SAS Procedures Guide, Version 6, Third Edition, 1990. SAS Institute Inc., Cary, NC.

SAS Institute Inc., SAS Macro Language: Reference, First Edition, 1990. SAS Institute Inc., Cary, NC.

Liu, Frank (1997), "A Standard Adverse Event Summary Report Generation," Proceedings of the twenty-second Annual SAS Users Group International Conference, 22, 1138-1144.

## ACKNOWLEDGMENT

SAS is a registered trademark of SAS Institute Inc..

The authors would like to thank Ischemia Research and Education Foundation for the data, and providing funding and support for this paper, and the helpful suggestions of Howard Zhong, consultant of Trilogy Consulting Corporation.

## Authors Contact

Amy C. Young (acy@crucis.iref.org)  
Sharon X. Zhou (sxz@crucis.iref.org)

Ischemia Research and Education Foundation  
250 Executive Park Blvd., suite 3400  
San Francisco, CA 94134-3306

```

*****
* PROGRAM: sugi24.sas
* PROGRAMMER: A. Young
* STUDY: SUGI-24
* INITIATION DATE: August 18, 1998
* DESCRIPTION: SAS CODE TO DEMONSTRATE USING SAS MACRO TO INCLUDE
* STATISTICS OUTPUT IN CLINICAL SUMMARY TABLE
*****

options center ps=58 errorabend mrecall mautosource mprint noovp
sasautos=(!SASROOT/sasautos;/users/analysis/acy/MACRO);

/* USER-DEFINED MACRO VARIABLES */
%LET GVAR = GRPCODE; * GROUP VARIABLE;
%LET INDATA = sugi24; * SAS DATA FOR MACRO PROGRAM -- REQUIRED;
%LET VARNUM = 5; * TOTAL NUMBER OF VARIABLES TO USE -- REQUIRED;

/* ASSIGN OUTPUT FILE LOCATION & FILE NAME */
FILENAME OUTFILE "/users/analysis/acy/sugi/sugi24.tab ";

/* ASSIGN TREATMENT GROUP FORMAT */
PROC FORMAT;
VALUE TRTFMT
1 = 'DOSE A'
2 = 'DOSE B'
3 = 'DOSE C'
4 = 'DOSE D'
5 = 'TOTAL' ;
RUN;

=====
* SAS PROGRAM DEVELOPMENT: PREPARATION OF ANALYSIS DATA SET
===== ;
/* OBTAIN INFORMATION FOR SUMMARY TABLE FROM CLINICAL TRIAL DATA.
NOTE: SUMMARY VARIABLES NAMES ARE IN ALPHABETICAL ORDER TO
CORRESPOND TO THE DESIRED ROW ORDER IN OUTPUT TABLE */

PROC SORT DATA=SUGI.SUGI24 (KEEP=ID GRPCODE CORETEMP CPLEGIA
XCOCODE4 XCOCODE5 XCOCODE7
OUT=TEMP1 (RENAME=(XCOCODE4=XAGRADE XCOCODE5=XRGRADE
XCOCODE7=XTALLOI)));

BY ID;
RUN;

DATA &INDATA;
SET TEMP1;
BY ID;
OUTPUT;
GRPCODE =5; * ASSIGN GROUP CODE 5 (=total number of treatments + 1) TO
CALCULATE THE "TOTAL";

OUTPUT;
RUN;

PROC SORT DATA=&INDATA;
BY &GVAR ;
RUN ;

=====
* GENERATING SUMMARY STATISTICS :
* STEP 1 -- ASSIGN MACRO VARIABLES TO EACH LISTED CATEGORIES,
* VARIABLE NAMES ARE CREATED IN ALPHABETICAL ORDER FOR LISTING
* PURPOSE.
===== ;

%LET VAR1 = CORETEMP; * VARIABLE NAME;
%LET VARTYP1 = CON; * TYPE (CON=CONTINUE, CAT=CATEGORICAL);
%LET VARFMT1 = ; * FORMAT IF ANY - DO NOT INCLUDE ".*";
%LET VARLAB1 = %STR(LOWEST CORE TEMPERATURE ON CPB); * LABEL ;

%LET VAR2 = CPLEGIA;
%LET VARTYP2 = CON;
%LET VARFMT2 = ;
%LET VARLAB2 = %STR(AMOUNT OF CARDIOPLEGIA (in mL));

%LET VAR3 = XAGRADE;
%LET VARTYP3 = CAT;
%LET VARFMT3 = ;
%LET VARLAB3 = %STR(ANTEGRADE) ;

%LET VAR4 = XRGRADE;
%LET VARTYP4 = CAT;
%LET VARFMT4 = ;
%LET VARLAB4 = %STR(RETROGRADE) ;

%LET VAR5 = XTALLOI;
%LET VARTYP5 = CAT;
%LET VARFMT5 = ;
%LET VARLAB5 = %STR(CRYSTALLOID) ;

=====
* GENERATING SUMMARY STATISTICS :
* STEP 2 -- MACRO TO GENERATE SUMMARY STATISTICS.
===== ;
/* MACRO: STATS
DESCR: GENERATE SUMMARY DATA OF ALL DEFINED SUMMARY VARIABLES */

%LET TRTNUM = ; * INITIALIZE TRTNUM MACRO VARIABLE;
%MACRO STATS(VAR=, /* SUMMARY VARIABLE */
VARFMT=, /* FORMAT OF VARIABLE, IF ANY */
VARTYP=, /* VAR TYPE: CON or CAT (CONTINUE or CATEGORICAL)
*/

VARLABEL= /* LABEL OF VARIABLE */);
%IF &TRTNUM= %THEN %DO; * RUN ONLY ONCE FOR FIRST TIME;

/* 1. FIND THE TOTAL NUMBER OF PATIENTS IN EACH GROUP */
PROC FREQ DATA=&INDATA;
TABLE &GVAR/OUT=TOTAL NOPRINT;
RUN;

DATA TOTAL;
SET TOTAL END=LAST;
IF LAST THEN CALL SYMPUT('TRTNUM',LEFT(PUT(_N_,3)));
RUN;

PROC TRANSPOSE DATA=TOTAL OUT=TTOTAL PREFIX=CNT;
ID &GVAR;
VAR COUNT;
RUN;

DATA ALL;
LENGTH VARNAME $8 VARLABEL $30 VARTYP $3 CATVAL $24;
SET TTOTAL(DROP=_NAME_ _LABEL_);
VARLABEL = 'NUMBER OF PATIENTS';
RUN;

%END;

/* 2. GENERATE SUMMARY STATISTICS */

*** CATEGORICAL VARIABLE ***;
%IF &VARTYP = CAT %THEN %DO;

PROC FREQ DATA=&INDATA;
TABLE &GVAR/OUT=FREQ1(DROP=PERCENT) NOPRINT;
TABLE &GVAR*&VAR/OUT=FREQ2(DROP=PERCENT) NOPRINT;
RUN;

DATA FREQS;
MERGE FREQ1(RENAME=(COUNT=TOTAL)) FREQ2;
BY &GVAR;
PERCENT = ( COUNT / TOTAL ) * 100 ;
RUN;

PROC SORT DATA=FREQS; BY &VAR &GVAR ; RUN;

PROC TRANSPOSE DATA=FREQS OUT=COUNTS PREFIX=CNT;
BY &VAR ; ID &GVAR;
VAR COUNT;
RUN;

PROC TRANSPOSE DATA=FREQS OUT=PERCENTS PREFIX=PCT;
BY &VAR ; ID &GVAR;
VAR PERCENT;
RUN;

DATA &VAR ;
LENGTH VARNAME $8 VARLABEL $30 VARTYP $3 CATVAL $24;
MERGE COUNTS(KEEP=&VAR CNT1-CNT&TRTNUM)
PERCENTS(KEEP=&VAR PCT1-PCT&TRTNUM);
BY &VAR ;
VARNAME = "&VAR"; VARLABEL = "&VARLABEL"; VARTYP = "&VARTYP";
%IF &VARFMT = %THEN %DO;
CATVAL = &VAR ;
%END;
%ELSE %DO;
CATVAL = PUT(&VAR,&VARFMT.);
%END;
DROP &VAR;
RUN;

DATA ALL; SET ALL &VAR ; RUN;

%END;

*** CONTINUOUS VARIABLE ***;
%IF &VARTYP = CON %THEN %DO;

PROC SORT DATA=&INDATA OUT=CONTINUE; BY &GVAR; RUN;

PROC UNIVARIATE DATA=CONTINUE NOPRINT;
BY &GVAR;
VAR &VAR ;
OUTPUT OUT=STATS MEAN=MEAN MIN=MIN MAX=MAX STD=STD N=COUNT
NMISS=MISS MEDIAN=MEDIAN;
RUN;

DATA &VAR (KEEP=VARNAME VARLABEL VARTYP CATVAL
MEAN1-MEAN&TRTNUM MIN1-MIN&TRTNUM MAX1-MAX&TRTNUM
STD1-STD&TRTNUM CNT1-CNT&TRTNUM MISCNT1-MISCNT&TRTNUM
MEDIAN1-MEDIAN&TRTNUM) ;
LENGTH VARNAME $8 VARLABEL $30 VARTYP $3 CATVAL $24;
ARRAY _MEAN{&TRTNUM} MEAN1-MEAN&TRTNUM;
ARRAY _MEDIAN{&TRTNUM} MEDIAN1-MEDIAN&TRTNUM;
ARRAY _MIN{&TRTNUM} MIN1-MIN&TRTNUM;
ARRAY _MAX{&TRTNUM} MAX1-MAX&TRTNUM;
ARRAY _STD{&TRTNUM} STD1-STD&TRTNUM;
ARRAY _NUM{&TRTNUM} CNT1-CNT&TRTNUM;
ARRAY _MISS{&TRTNUM} MISCNT1-MISCNT&TRTNUM;
VARNAME = "&VAR"; VARLABEL = "&VARLABEL";

```

```

VARTYP = "&VARTYP"; CATVAL = '';

DO I=1 TO &TRTNUM;
  SET STATS END=LAST;
  _MEAN(I) = MEAN;          _MEDIAN(I) = MEDIAN;
  _MIN(I) = MIN;           _MAX(I) = MAX;
  _STD(I) = STD;          _NUM(I) = COUNT;
  _MISS(I) = MISS;

  IF LAST THEN RETURN;
END;
RUN;

DATA ALL; SET ALL &VAR; RUN;

%END;
%MEND;

/* RUN STATS MACRO WITH USER DEFINED MACRO VARIABLES */
%MACRO READY;
%DO I = 1 %TO &VARNUM;
  %STATS(VAR=&&VAR&I,VARFMT=&&VARFMT&I,
  VARTYP=&&VARTYP&I,VARLABEL=&&VARLAB&I);
%END;
%MEND;

%READY;

PROC SORT DATA=ALL; BY VARNAME; RUN;

=====
* PRODUCING THE SUMMARY TABLE
===== ;

/* SETUP SUMMARY TABLE FORMAT */

%LET FOOTHTG=22; * NUMBER OF LINES IN FOOTER ;
%LET SPACING=1; * SPACE BETWEEN FIELDS ;
%LET FNUM=%EVAL(2 + &TRTNUM); * # OF FIELDS ;
%LET FW1=1; * WIDTH OF DEMOGRAPHIC VARIABLE;
%LET FW2=30; * WIDTH OF CATEGORY FIELD ;

/* ASSIGN COLUMN WIDTH FOR EACH TREATMENT GROUP */
%MACRO FINDGRP;
%DO I = 3 %TO %EVAL(&TRTNUM + 2);
  %GLOBAL FW&I;
  %LET FW&I = %EVAL(85/&TRTNUM);
%END;
%MEND;
%FINDGRP;

**** ASSIGN COLUMN FORMAT ****;
%MACRO SETCOLS;
%DO J=1 %TO &FNUM;
  %GLOBAL C&J;
%END;
%LET C1=10;
%DO J=2 %TO &FNUM;
  %LET JMINUS1=%EVAL(&J-1);
  %LET C&J=%EVAL(&&C&JMINUS1 + &&FW&JMINUS1 + &&SPACING);
  %PUT FIELD &J BEGINS AT COL &&C&J; * DEBUG LINE *;
%END;
%MEND SETCOLS;

%SETCOLS;

%LET LASTCOL=%EVAL(&&C&FNUM + &&FW&FNUM - &C1);
%PUT LAST COLUMN OF TABLE: &LASTCOL; * DEBUG LINE ;
%LET CENTER=%EVAL( &C1 + &LASTCOL/2 );
%PUT CENTER=&CENTER;

OPTIONS MISSING='';

=====
* GENERATE THE TABLE
===== ;

%MACRO MACRUN;
DATA _NULL_;
  LENGTH PPP1-PPP&TRTNUM $20;
  SET ALL END=EOF;
  BY VARNAME;
  FILE OUTFILE PRINT NOTITLES LINESIZE=133 PAGESIZE=60 N=PS
  LINESLEFT=LL;
  RETAIN NEXTPAGE 1 CONTINUE 0 MINUS1 -1;

  IF ( NEXTPAGE ) THEN LINK TITLES;

  ARRAY _PCT{&TRTNUM} PCT1-PCT&TRTNUM;
  ARRAY _CNT{&TRTNUM} CNT1-CNT&TRTNUM;
  ARRAY _PPP{&TRTNUM} PPP1-PPP&TRTNUM;

/* PUT OBSERVATION: (MODIFY TO FIT SPECIFIC STUDY) */

  IF FIRST.VARNAME THEN DO;
    PUT / @&C1 VARLABEL @;
  END;

  IF VARTYP = 'CON' THEN DO;
    PUT / @&C2 'MISSING'
    %DO I = 1 %TO &TRTNUM ;
      %LET J = %EVAL(&I+2);
      @&&C&J MISCNT&I 3.
    %END;
    / @&C2 'MEAN (S.D.)'
    %DO I = 1 %TO &TRTNUM ;
      %LET J = %EVAL(&I+2);
      @&&C&J MEAN&I 7.2 ' (' STD&I 6.1 ' )'
    %END;
    / @&C2 'MEDIAN'
    %DO I = 1 %TO &TRTNUM ;
      %LET J = %EVAL(&I+2);
      @&&C&J MEDIAN&I 6.1
    %END;
    / @&C2 'RANGE'
    %DO I = 1 %TO &TRTNUM ;
      %LET J = %EVAL(&I+2);
      @&&C&J MIN&I 6.1 ' TO ' MAX&I 6.1
    %END; ;
  END;

ELSE IF VARTYP = 'CAT' THEN DO;
  DO I = 1 TO &TRTNUM;
    IF _CNT(I) NE . THEN DO;
      IF _PCT(I) NE 100 THEN
        _PPP(I) = TRIM(PUT(_CNT(I),3.)) || '(' || TRIM(PUT(_PCT(I),4.1)) ||
        '%)';
      ELSE _PPP(I) = TRIM(PUT(_CNT(I),3.)) || '(100%)';
    END;
    ELSE _PPP(I) = ' -';
  END;
  PUT;
  %DO I = 1 %TO &TRTNUM ;
    %LET J = %EVAL(&I+2);
    IF _CNT(&I) < 10 THEN PUT @&C2 CATVAL @&&C&J+2 PPP&I @;
    ELSE PUT @&C2 CATVAL @&&C&J+1 PPP&I @;
  %END;
  IF LAST.VARNAME THEN PUT;
END;

ELSE IF VARTYP = '' THEN DO;
  PUT
  %DO I = 1 %TO &TRTNUM ;
    %LET J = %EVAL(&I+2);
    @&&C&J CNT&I 3.
  %END; ;
END;

** PUT FOOTER, RESET NEXTPAGE **;
IF EOF OR (LL < &FOOTHTG+1) THEN DO;
  LINK FOOTER; CONTINUE = 1;
END;
ELSE CONTINUE = 0;
RETURN;

*** PUT TITLES ****;
TITLES:
  NEXTPAGE=0;
  %DO I = 1 %TO &TRTNUM ;
    G&I = PUT(&I, TRTFMT.);
  %END;

** FIND TITLE COLUMN **;
TITL1COL=&CENTER - LENGTH(TITL1)/2;
TITL2COL=&CENTER - LENGTH(TITL2)/2;

** PUT TITLE & LABEL **;

PUT /// @&C1 "SUGL 24"
  @TITL1COL "CARDIOPULMONARY BYPASS – CARDIOPLEGIA SUMMARY"
  // @TITL2COL "-- SAMPLE OUTPUT FOR SUGI24 --"

  / @&C1 &LASTCOL*-'
  // @&C1 'VARIABLE'
  %DO I = 1 %TO &TRTNUM ;
    %LET J = %EVAL(&I+2);
    @&&C&J G&I /'*** 'GROUP '****/
  %END;
  / @&C1 &LASTCOL*-' ;

RETURN;

** PUT FOOTNOTES **;
FOOTER:
  DO WHILE (LL > &FOOTHTG);
    PUT;
  END;

  NEXTPAGE=1; RETURN;
RUN;
%MEND;

%MACRUN;

=====
* PROGRAM DEVELOPMENT ENDS HERE
===== ;

```

SUGI\_24

 CARDIOPULMONARY BYPASS -- CARDIOPLEGIA SUMMARY  
 -- SAMPLE OUTPUT FOR SUGI24 --

VARIABLE	DOSE A	DOSE B	DOSE C	DOSE D	TOTAL
NUMBER OF PATIENTS	12	12	14	11	49
LOWEST CORE TEMPERATURE ON CPB					
MISSING	0	0	0	0	0
MEAN (S.D.)	31.65 ( 2.5)	31.69 ( 2.7)	31.39 ( 2.1)	32.16 ( 1.7)	31.70 ( 2.2)
MEDIAN	31.5	31.6	31.1	32.0	31.5
RANGE	28.0 TO 35.2	27.9 TO 35.4	28.0 TO 34.0	30.0 TO 35.2	27.9 TO 35.4
AMOUNT OF CARDIOPLEGIA (in mL)					
MISSING	0	0	0	0	0
MEAN (S.D.)	763.33 ( 548.1)	1295.75 (2218.9)	1816.07 (2556.6)	999.64 ( 791.3)	1247.55 (1807.0)
MEDIAN	605.0	570.0	667.5	600.0	600.0
RANGE	300.0 TO 2250.0	220.0 TO 8100.0	160.0 TO 9930.0	340.0 TO 2400.0	160.0 TO 9930.0
ANTEGRADE					
.	-	2 (16.7%)	1 ( 7.1%)	1 ( 9.1%)	4 ( 8.2%)
Yes	12 (100%)	10 (83.3%)	13 (92.9%)	10 (90.9%)	45 (91.8%)
RETROGRADE					
.	7 (58.3%)	5 (41.7%)	7 (50.0%)	6 (54.5%)	25 (51.0%)
Yes	5 (41.7%)	7 (58.3%)	7 (50.0%)	5 (45.5%)	24 (49.0%)
CRYSTALLOID					
.	4 (33.3%)	2 (16.7%)	4 (28.6%)	4 (36.4%)	14 (28.6%)
Yes	8 (66.7%)	10 (83.3%)	10 (71.4%)	7 (63.6%)	35 (71.4%)